Differentiation

J. Gerlach
November 2010

D and diff

The limit definition of the derivative is covered in the *Limit* tutorial. Here we look for direct ways to calculate derivatives. Maple has two commands for this task. One is the **D** operator, the other is the **diff** command. These commands behave quite differently, and we shall illustrate both for the example $f(x) = \frac{1}{1+x^2}$.

The Use of D

D acts on functions, and we have to define f(x) first.

$f := x \to \frac{1}{1+x^2};$

$$x \to \frac{1}{1+x^2} \quad \textbf{(1.1.1)}$$

Now taking the derivative is straightforward, just enter

$\mathrm{D}(f)(x);$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.1.2)}$$

If you type

$\mathrm{D}(f);$

$$x \to -\frac{2x}{(1+x^2)^2} \quad \textbf{(1.1.3)}$$

maple returns a function instead. This all goes back to distinction between a function f and a formula $f(x)$ for the output, the latter being an expression. In short, $\mathrm{D}(f)$ stands for f', and therefore $\mathrm{D}(f)(x)$ represents f'(x). If you want to evaluate the derivative at a specified value, say you want the slope of our function at $x = 3$ you would enter

$\mathrm{D}(f)(3);$

$$-\frac{3}{50} \quad \textbf{(1.1.4)}$$

The Use of diff

diff acts on expressions. You can enter the expression directly, and you do not have to define the function in advance. In our example we have the options

$\mathit{diff}\left(\frac{1}{1+x^2}, x\right);$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.2.1)}$$

or (the definition of f is still in effect)
$diff(f(x), x);$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.2.2)}$$

or in two steps
$y := \frac{1}{1+x^2};$

$$\frac{1}{1+x^2} \quad \textbf{(1.2.3)}$$

$diff(y, x);$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.2.4)}$$

In all cases the command has the form diff(expression, variable). The template on the left executes the diff command
$\frac{d}{dx} y;$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.2.5)}$$

Evaluation of the derivative at a point requires two steps (use $x=3$ again).
$diff(y, x);$

$$-\frac{2x}{(1+x^2)^2} \quad \textbf{(1.2.6)}$$

$subs(x=3, \%);$

$$-\frac{3}{50} \quad \textbf{(1.2.7)}$$

Do not mix D and diff

The following results in nonsense (the definitions of f(x) and y are still in effect).
$D(y)(x);$

$$-\frac{2\,D(x)(x)\,x(x)}{(1+x(x)^2)^2} \quad \textbf{(1.3.1)}$$

$diff(f, x);$

$$0 \quad \textbf{(1.3.2)}$$

Clickable Math

Enter the expression and right-click.
$\frac{1}{1+x^2}$

$$\frac{1}{1+x^2} \quad \textbf{(1.4.1)}$$

$\xrightarrow{\text{differentiate w.r.t. x}}$

$$-\frac{2x}{(1+x^2)^2} \qquad (1.4.2)$$

evaluate at point →

$$-\frac{3}{50} \qquad (1.4.3)$$

More Examples

Problem: Find the derivative of $f(x) = x^2 e^{-x}$.

$f := x \rightarrow x^2 \cdot \exp(-x);$

$$x \rightarrow x^2 e^{-x} \qquad (1.5.1)$$

$D(f)(x);$

$$2\,x\,e^{-x} - x^2 e^{-x} \qquad (1.5.2)$$

Problem: Find the derivative of $f(x) = (2\,x+1)^9 (5\,x-2)^{12}$ and simplify the result

$f := x \rightarrow (2 \cdot x + 1)^9 \cdot (5 \cdot x - 2)^{12};$

$$x \rightarrow (2\,x+1)^9\,(5\,x-2)^{12} \qquad (1.5.3)$$

$D(f)(x);$

$$18\,(2\,x+1)^8\,(5\,x-2)^{12} + 60\,(2\,x+1)^9\,(5\,x-2)^{11} \qquad (1.5.4)$$

$factor(\%);$

$$6\,(4+35\,x)\,(2\,x+1)^8\,(5\,x-2)^{11} \qquad (1.5.5)$$

Problem: Find the slope of the function $f(x) = \sin^2(x)$ at the point where $x = \frac{5\,\pi}{3}$.

$f := x \rightarrow \sin(x)^2;$

$$x \rightarrow \sin(x)^2 \qquad (1.5.6)$$

$D(f)\left(\frac{5 \cdot \pi}{3}\right);$

$$-\frac{1}{2}\sqrt{3} \qquad (1.5.7)$$

Higher Order Derivatives

D and diff

Let's say we want to compute the 5th derivative of the function $f(x) = x^2 \cdot e^{-x}$. Using the **D** operator, we define the function and then use a modified D command, namely **(D@@n)**, where n is the order of the derivative.

$f := x \rightarrow x^2 \cdot \exp(-x);$

$$x \rightarrow x^2 e^{-x} \qquad (2.1.1)$$

$(D@@5)(f)(x);$

$$-20\,e^{-x} + 10\,x\,e^{-x} - x^2\,e^{-x} \quad \textbf{(2.1.2)}$$

There appears to be a common factor

factor(%);

$$-e^{-x}\,(20 - 10\,x + x^2) \quad \textbf{(2.1.3)}$$

With the **diff** command, we repeatedly enter x, once of each derivative.

$diff(x^2 \cdot e^{-x}, x, x, x, x, x);$

$$-20\,e^{-x} + 10\,x\,e^{-x} - x^2\,e^{-x} \quad \textbf{(2.1.4)}$$

This can get rather lengthy if you need a real high order derivative. Fortunately, there is a fix for this problem using x$n.

$diff(x^2 \cdot e^{-x}, x\$5);$

$$-20\,e^{-x} + 10\,x\,e^{-x} - x^2\,e^{-x} \quad \textbf{(2.1.5)}$$

Higher Order Derivatives at a Point

Evaluation of a higher order derivative at a specified point is again a little simpler with the **D** operator. Let's compute the eight derivative of $f(x) = x \cdot \sin(x)$ and evaluate it at π. First we will use the **D** operator.

$f := x \rightarrow x \cdot \sin(x);$

$$x \rightarrow x \sin(x) \quad \textbf{(2.2.1)}$$

$(D@@8)(f)(\pi);$

$$8 \quad \textbf{(2.2.2)}$$

With the diff command we proceed as follows.

$diff(x \cdot \sin(x), x\$8);$

$$-8\cos(x) + x\sin(x) \quad \textbf{(2.2.3)}$$

$subs(x = \mathrm{Pi}, \%);$

$$-8\cos(\pi) + \pi\sin(\pi) \quad \textbf{(2.2.4)}$$

value(%);

$$8 \quad \textbf{(2.2.5)}$$

More Examples

Problem: Compute f'(x) and f''(x) when $f(x) = (x^2 + 1) \cdot (e^x + 1)$.

$diff((x^2 + 1) \cdot (\exp(x) + 1), x);$

$$2\,x\,(e^x + 1) + (x^2 + 1)\,e^x \quad \textbf{(2.3.1)}$$

simplify(%); # *first derivative*

$$2\,x\,e^x + 2\,x + e^x\,x^2 + e^x \quad \textbf{(2.3.2)}$$

diff(%, x); # *second derivative*

$$3\,e^x + 2 + 4\,x\,e^x + e^x\,x^2 \quad \textbf{(2.3.3)}$$

or

$f := x \rightarrow (x^2 + 1) \cdot (\exp(x) + 1);$

$$x \rightarrow (x^2 + 1)\,(e^x + 1) \quad \textbf{(2.3.4)}$$

simplify(D(*f*) (*x*));

$$2\,x\,e^x + 2\,x + e^x x^2 + e^x \quad \textbf{(2.3.5)}$$

simplify((D@@2)(*f*) (*x*));

$$3\,e^x + 2 + 4\,x\,e^x + e^x x^2 \quad \textbf{(2.3.6)}$$

Problem: Compute the 15th derivative of $f(x) = e^{-2x}$.

diff(exp(−2·*x*), *x*$15);

$$-32768\,e^{-2x} \quad \textbf{(2.3.7)}$$

More Examples

A Function and its Derivative in a Common Graph

Problem: Sketch the function $f(x) = \sin(x - \sin(x))$ and its derivative in a common figure.

f := *x* → sin(*x* − sin(*x*)); *# Definition of the function*

$$x \rightarrow \sin(x - \sin(x)) \quad \textbf{(3.1.1)}$$

plot([*f*, D(*f*)], −2 ..8, *color* = [*red*, *blue*], *thickness* = [2, 1]);



The graph of the function is in red, its derivative in blue. It is evident that the blue curve measures the slope of the red curve.

Function with a Tangent Line in a Common Figure

Problem: Graph the function $f(x) = \sqrt{x}$ and its tangent line at x=4 in a common figure.

f := *x* → sqrt(*x*); *# define the function f*

$$x \rightarrow \sqrt{x} \quad \textbf{(3.2.1)}$$

a := 4; *# select the point*

$$4 \quad \textbf{(3.2.2)}$$

y := *f*(*a*) + D(*f*) (*a*)·(*x* − *a*); *# calculate the tangent line*

$$1 + \frac{1}{4}\,x \quad \textbf{(3.2.3)}$$

plot([*f*(*x*), *y*], *x* = 0 ..8, *color* = [*blue*, *black*]); *# graph both*



Find Horizontal Tangent Lines

Problem: Find all points in the interval $[0, 2\pi]$ for which the graph of $f(x) = \sin(x - \sin(x))$ has a horizontal tangent line.
In order to solve the problem, we need to calculate the derivative , set it to zero, and solve the resulting equation for x.

$f := x \rightarrow \sin(x - \sin(x));$

$$x \rightarrow \sin(x - \sin(x))$$

$solve(\ D(f)\ (x) = 0\ , x);$

$$RootOf(2\ \_Z - 2\sin(\_Z) - \pi), 0 \quad \textbf{(3.3.2)}$$

Here we tried to solve f'(x) = 0, but maple did not get a reasonable answer. Let's look at the graph of f.
$plot(\ f(x), x = 0\,..2\cdot \text{Pi});$



The function appears to have zero slope at the endpoints and at two more points in the inside.
Let's see what we get from fsolve, and label the points.
$a := fsolve(\ D(f)\ (x) = 0, x);$

$$2.309881460 \quad \textbf{(3.3.3)}$$

$b := fsolve(D(f)\ (x) = 0, x = 4);$

$$3.973303847 \quad \textbf{(3.3.4)}$$

$c := fsolve(D(f)\ (x) = 0, x = 0);$

$$0. \quad \textbf{(3.3.5)}$$

$d := fsolve(\mathrm{D}(f)(x) = 0, x = 6.2);$

$$6.283185306 \quad \textbf{(3.3.6)}$$

Here are the coordinates of the points:

$(c, f(c)); (a, f(a)); (b, f(b)); (d, f(d));$

$$0., 0.$$

$$2.309881460, 1.$$

$$3.973303847, -1.$$

$$6.283185306, -1.795864770\ 10^{-10} \quad \textbf{(3.3.7)}$$

Implicit Differentiation

The implicitdiff Command

Here we take the derivative on both sides of an equation. We need to set up the equation, and decide on the dependent and the independent variable. We will use $x^3 + y^3 = \frac{9\,xy}{2}$ as an example. Note that the point (1,2) lies on this curve, and we shall determine the slope there. First we define the equation for future reference.

restart;

$Eqn := x^3 + y^3 = \frac{9 \cdot x \cdot y}{2};$

$$x^3 + y^3 = \frac{9}{2}\,x\,y \quad \textbf{(4.1.1)}$$

First we confirm that the point (1,2) lies on the curve. To do so, we substitute the coordinates into the equation.

$subs(\{x = 1, y = 2\}, Eqn);$

$$9 = 9 \quad \textbf{(4.1.2)}$$

It worked!. Now we use the command implicitdiff. It has the structure implicitdiff(equation, dependent variable, independent variable). In our case we have

$implicitdiff(\ Eqn, y, x);$

$$\frac{-2\,x^2 + 3\,y}{2\,y^2 - 3\,x} \quad \textbf{(4.1.3)}$$

This is $\frac{dy}{dx}$, representing the slope. If we substitute the coordinates of our point, we obtain

$m := subs(\{x = 1, y = 2\}, \%);$

$$\frac{4}{5} \quad \textbf{(4.1.4)}$$

Let's illustrate the result graphically. In order to graph the equation, we need a more sophisticated plotting command, and we have to load the plots package. Now the command implicitplot will do the desired task.

$with(plots):$

$implicitplot(\ Eqn, x = -1\,..3, y = -1\,..3);$



Now let's include the tangent line as well (I used the point-slope form without solving for y first).

implicitplot([*Eqn*, $y-2=m\cdot(x-1)$], $x=-1..3$, $y=-1..3$, *scaling* = *constrained*, *color* = [*red*, *blue*], *numpoints* = 2500);



More Examples

Problem: Find dy/dx from the equation y = sin(xy).

implicitdiff($y=\sin(x\cdot y)$, *y*, *x*);

$$-\frac{\cos(x\,y)\,y}{\cos(x\,y)\,x-1} \quad \textbf{(4.2.1)}$$

Problem: What is the slope of the curve given by $\sqrt{x}+\sqrt{y}=5$ at the point (9,4)?

Equation := sqrt(x) + sqrt(y) = 5;

$$\sqrt{x}+\sqrt{y}=5 \quad \textbf{(4.2.2)}$$

implicitdiff(*Equation*, *y*, *x*);

$$-\frac{\sqrt{y}}{\sqrt{x}} \quad \textbf{(4.2.3)}$$

subs({$x=9$, $y=4$}, %);

$$-\frac{1}{9}\sqrt{4}\,\sqrt{9} \quad \textbf{(4.2.4)}$$

simplify(%);

$$-\frac{2}{3} \tag{4.2.5}$$

Problem: Find the points with horizontal tangents on the curve $(x^2+y^2)^2 = 4\,x^2 y$.
First define the curve as C

$C := (x^2+y^2)^2 = 4\cdot x^2\cdot y;$

$$(x^2+y^2)^2 = 4\,x^2\,y \tag{4.2.6}$$

Derivative by implicit differentiation

$dC := implicitdiff(\ C\ , y, x);$

$$-\frac{x\,(x^2+y^2-2\,y)}{x^2\,y+y^3-x^2} \tag{4.2.7}$$

Now set the derivative to zero.

$solve(\ dC = 0, x);$

$$0, \sqrt{-y^2+2\,y}, -\sqrt{-y^2+2\,y} \tag{4.2.8}$$

Clearly, one solution occurs when x=y=0. The other two occur when $x=\sqrt{2\,y-y^2}$ or the negative of it. In order to identify the points, we need to substitute these values into the original equation. As an alternative, we could solve the original equation along with dC=0 simultaneously.

$solve(\ \{C\ , dC = 0\}, \{x, y\});$

$$\{x=-1, y=1\}, \{x=1, y=1\} \tag{4.2.9}$$

Now we have the solutions (-1,1) and (1,1). Maple probably ignored the origin, because the derivative is undefined at this point.